



MUSICAL STINGS

BY JOEY LATIMER

I recently playtested a computer game a friend of mine wrote. The game is called *Incredible Exploder*, and it took me days to get good at it. When I finally got close to winning, I started dreaming of my grand finale, my moment of glory. Would there be a flourish of trumpets? A synthesized symphony? A victory serenade?



I didn't have long to wait before the answer came. I ducked left, dove right, and scurried off the screen to victory. Aargh!!

After several seconds, I peeked out from behind my hands. Printed in the center of the screen were two tiny words: YOU WON. Needless to say, I was bummed. What kind of reward was that?!

If good music is what your computer games are missing, you've come to the right place. This month's Microtones brings you *Musical Stings*—short music routines you can use to add drama to your programs. We've got a "winner sting," which plays a victory ditty, and a "loser sting," which plays a dreary tune. ADAM and Apple owners will be glad to see we've specially created music programs for them! There's also a version for the CoCo, Atari, TI, Commodore 64, and VIC-20. (Atari, TI, Commodore 64, and VIC-20 owners can look in Hacker Heaven Programs for other music stuff.)

Musical Stings consists of three sections. The first part of the program DIMensions two arrays, N and D (short for note and duration), and fills them with our music data. (Machine language is necessary to

make music on ADAM and Apple computers, so the first section of those versions also contains a loop that reads in a machine-language routine and stores it in free memory.) The second and third parts of the program are the winner and loser stings—routines that play different parts of the music contained in the arrays.

Although *Musical Stings* will run correctly just as it appears, it's easy to make it part of your own computer masterpiece. First, put the initial routine and DATA statements near the beginning of your listing. Then, put the winner sting and loser sting routines wherever you want a happy or a sad tune to play. If your program already has DATA statements of its own, you may have to rearrange a few lines so that our data is read in correctly. If your program contains a RESTORE statement, you'll probably have to change other elements, too. Don't hesitate to change our line numbering, but be careful!

On this note, ADAM and Apple owners should be very careful entering DATA—there's machine code there, and a single typo could bomb your system! Remember to SAVE a copy of your program to disk or tape before testing.

If our variable or array names conflict with those already in your program, you can change them too. Just make sure that you change them consistently throughout the initial routine, the winner sting, and the loser sting.

JOEY LATIMER is K-POWER's associate technical editor. He's a musician and a programming pro.



APPLE/MUSICAL STINGS

II plus, IIe, or IIc • 16K RAM

```

9 REM --INITIAL ROUTINE; PUT NEAR START OF PROGRAM--
10 DIM N(30),D(30)
20 FOR X = 1 TO 30
30 READ N(X),D(X)
40 NEXT X
50 FOR X = 0 TO 28
60 READ ML
70 POKE 768 + X,ML
80 NEXT X
99 REM --MUSIC DATA--
100 DATA 215,50,160,50,168,25,160,25,142,50,160,25
110 DATA 142,25,125,50,142,25,125,25,116,50,125,25
120 DATA 116,25,105,50,116,25,125,25,116,25,125,25
130 DATA 142,50,160,50,215,50,160,50,160,120,215,40
140 DATA 225,40,215,40,202,120,215,240,168,120,160,120
149 REM --MACHINE CODE. TYPE THIS IN CAREFULLY!--
150 DATA 165,8,74,133,10,164,8,173,48,192,136,234
    
```



```

160 DATA 234,208,251,165,7,56,229,10,133,7,176,237
170 DATA 198,6,208,233,96
199 REM --THIS IS THE WINNER STING--
200 FOR X = 1 TO 22
210 POKE 8,N(X)
220 POKE 6,D(X)
230 CALL 768
240 NEXT X
299 REM --THIS IS THE LOSER STING--
300 FOR X = 23 TO 30
310 POKE 8,N(X)
320 POKE 6,D(X)
330 CALL 768
340 NEXT X

```



ATARI/MUSICAL STINGS

400, 600XL, 800, or 800XL • 16K RAM

```

9 REM --INITIAL ROUTINE; PUT NEAR START OF PROGRAM--
10 DIM N(30),D(30)
20 FOR X=1 TO 30
30 READ NT,DUR
40 N(X)=NT
50 D(X)=DUR
60 NEXT X
69 REM --MUSIC DATA--
70 DATA 108,85,81,85,85,42,81,42,72,85,81,42
80 DATA 72,42,64,85,72,42,64,42,60,85,64,42
90 DATA 60,42,53,85,60,42,64,42,60,42,64,42,72,85
100 DATA 81,85,108,85,81,85,81,147,108,49,114,49
110 DATA 108,49,102,147,108,343,85,147,81,123
199 REM --THIS IS THE WINNER STING--
200 FOR X=1 TO 22
210 SOUND 0,N(X),10,10
220 FOR D=1 TO D(X)
230 NEXT D
240 NEXT X
250 SOUND 0,0,0,0
299 REM --THIS IS THE LOSER STING--
300 FOR X=23 TO 30
310 SOUND 0,N(X),10,10
320 FOR D=1 TO D(X)
330 NEXT D
340 NEXT X
350 SOUND 0,0,0,0

```



COLECO/MUSICAL STINGS

ADAM • 80K RAM • TV, or monitor with audio plug connected

```

9 REM --INITIAL ROUTINE; PUT AT START OF PROGRAM--
10 LOMEM: 29000
20 DIM n(30),d(30)
30 FOR x = 1 TO 30
40 READ n(x),d(x)
50 NEXT x
60 FOR x = 28000 TO 28005

```

```

70 READ ml
80 POKE x,ml
90 NEXT x
99 REM --MUSIC DATA--
100 DATA 35,200,26,200,28,100,26,100,23,200,26,100
110 DATA 23,100,21,200,23,100,21,100,20,200,21,100
120 DATA 20,100,17,200,20,100,21,100,20,100,21,100
130 DATA 23,200,26,200,35,200,26,200,26,300,35,100
140 DATA 37,100,35,100,33,300,35,600,28,300,26,300
149 REM --MACHINE CODE; TYPE THIS IN CAREFULLY!--
150 DATA 58,102,109,211,255,201
199 REM --THIS IS THE WINNER STING--
200 FOR x = 1 TO 22
210 POKE 28006,128:CALL 28000
220 POKE 28006,n(x):CALL 28000
230 POKE 28006,144:CALL 28000
240 FOR d = 1 TO d(x)
250 NEXT d
260 NEXT x
270 POKE 28006,159:CALL 28000
299 REM --THIS IS THE LOSER STING--
300 FOR x = 23 TO 30
310 POKE 28006,128:CALL 28000
320 POKE 28006,n(x):CALL 28000
330 POKE 28006,144:CALL 28000
340 FOR d = 1 TO d(x)
350 NEXT d
360 NEXT x
370 POKE 28006,159:CALL 28000

```

Special thanks to Ramsey J. Benson, coauthor of *ADAM's Companion* (Avon Books, 1984, \$9.95).



COMMODORE/MUSICAL STINGS

Commodore 64 • 64K RAM

```

9 REM --INITIAL ROUTINE; PUT NEAR START OF PROGRAM--
10 DIM N(30,2),D(30)
20 S=54272
30 FOR E=S TO S+28
40 POKE E,0
50 FOR X=1 TO 30
60 READ N(X,1),N(X,2),D(X)
70 NEXT X
80 POKE S+24,15
90 POKE S+5,39
100 POKE S+6,252
109 REM --MUSIC DATA--
110 DATA 18,209,200,25,30,200,23,181,100,25,30,100
120 DATA 28,49,200,25,30,100,28,49,100,31,165,200
130 DATA 28,49,100,31,165,100,33,135,200,31,165,100
140 DATA 33,135,100,37,162,200,33,135,100,31,165,100
150 DATA 33,135,100,31,165,100,28,49,200,25,30,200
160 DATA 18,209,200,25,30,200,25,30,300,18,209,100
170 DATA 17,195,100,18,209,100,19,239,300,18,209,600
180 DATA 23,181,300,25,30,200
199 REM --THIS IS THE WINNER STING--
200 FOR X=1 TO 22
210 POKE S+1,N(X,1)
220 POKE S,N(X,2)
230 POKE S+4,17
240 FOR DUR=1 TO D(X)
250 NEXT DUR
260 POKE S+4,16
270 NEXT X
299 REM --THIS IS THE LOSER STING--
300 FOR X=23 TO 30

```



```

310 POKE S+1,N(X,1)
320 POKE S,N(X,2)
330 POKE S+4,17
340 FOR DUR=1 TO D(X)
350 NEXT DUR
360 POKE S+4,16
370 NEXT X

```



COMMODORE/MUSICAL STINGS

VIC-20 • 5K RAM

```

9 REM --INITIAL ROUTINE; PUT NEAR START OF PROGRAM--
10 DIM N(33),D(33)
20 FOR X=1 TO 33
30 READ N(X),D(X)
40 NEXT X
50 POKE 36878,8
59 REM --MUSIC DATA--
60 DATA 201,192,215,192,212,96,215,96,219,192,215,96
70 DATA 219,96,223,192,219,96,223,96,225,192,223,96
80 DATA 225,96,228,192,225,96,223,96,225,96,223,96
90 DATA 219,192,215,192,201,192,215,192,0,0,215,288
100 DATA 201,96,199,96,201,96,203,288,201,384,0,288
110 DATA 212,288,215,240,0,0
199 REM --THIS IS THE WINNER STING--
200 FOR X=1 TO 23
210 POKE 36876,N(X)
220 FOR DUR=1 TO D(X)
230 NEXT DUR,X
299 REM --THIS IS THE LOSER STING--
300 FOR X=24 TO 33
310 POKE 36876,N(X)
320 FOR DUR=1 TO D(X)
330 NEXT DUR,X

```



RADIO SHACK/MUSICAL STINGS

TRS-80 Color Computer • 16K RAM

```

9 REM --INITIAL ROUTINE; PUT NEAR START OF PROGRAM--
10 DIM N(30),D(30)
20 FOR X=1 TO 30
30 READ N(X),D(X)
40 NEXT X
49 REM --MUSIC DATA--
50 DATA 108,6,147,6,140,3,147,3,159,6,147,3
60 DATA 159,3,170,6,159,3,170,3,176,6,170,3
70 DATA 176,3,185,6,176,3,170,3,176,3,170,3
80 DATA 159,6,147,6,108,6,147,6,147,10,108,3
90 DATA 99,3,108,3,117,10,108,23,140,10,147,8
99 REM --THIS IS THE WINNER STING--
100 FOR X=1 TO 22
110 SOUND N(X),D(X)
120 NEXT X
199 REM --THIS IS THE LOSER STING--
200 FOR X=23 TO 30
210 SOUND N(X),D(X)
220 NEXT X

```



TEXAS INSTRUMENTS/MUSICAL STINGS

TI-99/4A • 16K RAM

```

9 REM --INITIAL ROUTINE; PUT NEAR START OF PROGRAM--
10 DIM N(30)
20 DIM D(30)
30 FOR X=1 TO 30
40 READ N(X),D(X)
50 NEXT X
59 REM --MUSIC DATA--
60 DATA 295,300,392,300,370,150,392,150,440,300
70 DATA 392,150,440,150,494,300,440,150,494,150
80 DATA 523,300,494,150,523,150,587,300,523,150
90 DATA 494,150,523,150,494,150,440,300,392,300
100 DATA 294,300,392,300,392,600,294,200,277,200
110 DATA 294,200,311,600,294,1200,370,600,392,600
199 REM --THIS IS THE WINNER STING--
200 FOR X=1 TO 22
210 CALL SOUND(D(X),N(X),5)
220 NEXT X
299 REM --THIS IS THE LOSER STING--
300 FOR X=23 TO 30
310 CALL SOUND(D(X),N(X),5)
320 NEXT X

```

We'll Knock Your Socks Off For Five Bucks!

What can \$5 get you these days? EnTech's \$5 **Studio 64 Album** deal gives you a lot! You'll hear ten hot songs and a DJ who talks in an actual human voice. Watch the notes move across the screen in high resolution graphics. You'll also get a \$5 coupon good for EnTech's advanced music synthesizer for the Commodore 64, **Studio 64**. You can't beat a deal like that, and you can't beat the way **Studio 64** sounds! We guarantee it's better than **Music Construction Set** or **Musicalc**. Order your \$5 **Studio 64 Album** today!

Limited Time Offer



P.O. BOX 185, SUN VALLEY, CA 91353
(818) 768-6646



SAVE COD & CREDIT CHARGES BY
SENDING CHECK OR MONEY ORDER



Musical Pursuit: What music software company president plays accordion, bass guitar, piano, banjo, drums and congos, string bass, organ, mandolin, has fooled around with cello and violin, and plays a synthesizer?!

None other than **Ray Soular** of **EnTech**. Yes, that's the same EnTech which sponsors the First Annual Computer Song Writing Contest and produces a music program called **Studio 64**! (The deadline for The Computer Song Writing Contest, by the way, has been extended until Dec. 1. The prize is \$1,000 plus free recording time for the best song written on the Commodore 64 using EnTech's *Studio 64*. Mail to: Computer Song Contest, P.O. Box 185, Sun Valley, CA 91353.)

Ray, age 28, comes from a musical family. When he and his three brothers and one sister get together with their music-teacher pop, it's like one giant jam session! "Music is our family coat of arms," Ray says.

Having a jam session with your C 64 is what EnTech is all about. Ray explains, "We wanted to create a music package where the user could get the whole [of the Beatles'] *Sgt. Pepper* in there." The latest version of *Studio 64* is just out and has been improved. "It now has high-resolution graphics and 12 sample songs," says Ray. "Plus, you can now use sharps and flats and the control key functions."

Studio 64 sells for approximately \$40. And there's other big EnTech news—before Christmas look for a new **EnTech keyboard**!



EnTech president Ray Soular plays 10 instruments!

We got the beat! The drum beat, that is! PVI sent us the **Drum-Key**, an electronic drum synthesizer for the Apple. It uses digital recordings of 28 different percussion sounds—kick-bass, snare drum, four tom-toms, cymbals, cowbells, and tambourines. Drum-Key's features include multitrack recording, single-step editing, programmable

tempos, demo patterns and songs, and more. Want one? Call toll free: (800) 441-1003. \$139

Ian Burden (guitarist and keyboard player for the **Human League**) admitted in a recent interview that Human League had little experience playing instruments when they recorded their first hit record "Don't You Want Me (Baby)." But, thanks to computers, it didn't matter.



The Human League.

"You can write songs, you can think up tunes in your head, and if you can't play an instrument, then computers are there to enable you to do it," Ian told the *New York Daily News*. "So a lot of people who might have been restricted by not being able to play before, but who had good ideas, can do something."

But Ian says the group's reliance on computers also hurt it. "The first thing we ever did was to make a record. There hadn't been [enough] playing together as a group." And once their producer, Martin Rushent (whose computer smarts helped forge the League's unique sound) left the group, they decided to start playing traditional instruments.

What do most Silicon Valley hackers want to do with their computers? Make music! A computerized music course at Stanford University (in the heart of the Valley), had 10 times the number of applicants than it could take! (And, in the Valley, there are more computers per person than anywhere else.)

We just received a great magazine that covers the world of electronic music! It's called **RUG (Roland Users Group magazine)** and it has great articles about everything from **Duran Duran**, to high-school marching bands, to all the hottest music hardware. You can get a copy by writing to: Roland Corp. US, 7200 Dominion Circle, Los Angeles, CA 90040.




Illustration: Chris Reed



<p>PROGRAMS Page 26 <i>Word Twister</i> winners! Palindromes! Martians! Mozart! And a Drumulator?</p>	<p>PIXEL THAT! Page 32 Graphics programs for Apple, Atari, CoCo, and IBM. Learn to fly!</p>
<p>COMPUCOPIA Page 35 Short short programs: Encoder/Decoder Contest winners!</p>	<p>MICROTONES Page 37 <i>Musical Stings</i> for ADAM, Apple, Atari, CoCo, Commodore, and TI!</p>

Illustration: Jim Cherry III